

Stanislav Bartoň

The dynamics of the drum mower blade

In: Jan Chleboun and Pavel Kůs and Petr Příkryl and Miroslav Rozložník and Karel Segeth and Jakub Šístek and Tomáš Vejchodský (eds.): Programs and Algorithms of Numerical Mathematics, Proceedings of Seminar. Hejnice, June 24-29, 2018. Institute of Mathematics CAS, Prague, 2019. pp. 7–14.

Persistent URL: <http://dml.cz/dmlcz/703063>

Terms of use:

© Institute of Mathematics CAS, 2019

Institute of Mathematics of the Czech Academy of Sciences provides access to digitized documents strictly for personal use. Each copy of any part of this document must contain these *Terms of use*.



This document has been digitized, optimized for electronic delivery and stamped with digital signature within the project *DML-CZ: The Czech Digital Mathematics Library*
<http://dml.cz>

THE DYNAMICS OF THE DRUM MOWER BLADE

Stanislav Bartoň

Opole University of Technology
Prószkowska 76 Street, 45-758 Opole, Poland
s.barton@po.opole.pl

Abstract: The drum mower blade is freely rotatable around the fastening pin. During the operation of the mower, the centrifugal force and the resistance of the mowing material act on it. The presented article studies the effect of these forces on the behavior of the blade, in particular its oscillation around the steady state, depending on the properties of the cut material.

Keywords: momentum of force, momentum of inertia, nonlinear differential equation, Maple

MSC: 34A34, 68U20, 70B10

1. Introduction

To solve the problem, we choose the two-dimensional rectangular coordinate system. The direction of the x -axis is selected in the opposite direction to the speed of the mower. The origin C of this coordinate system is selected on the axis of rotation of the drum. R is the rotation radius of the mounting pin, the dimensions of the rectangular blade are a , and b . The radius of the mounting pin is r , the mounting pin is located on the axis of the blade and is at a distance Sb from the shorter side of the blade a , see Fig. 1, detailed description can be downloaded from [2].

The blade rotates with angular velocity ω counterclockwise, the mower moves in the negative direction of the x axis. The centrifugal force keeps the blade in the radial direction – the equilibrium position. Due to the cutting resistance, the blade is deflected by the angle ψ from the equilibrium position. If the blade is not in the equilibrium, the centrifugal force acts on it by a torque that tries to return it to equilibrium. The magnitude of the return torque depends on the angular velocity ψ , the dimensions and weight of the blade, and the angular velocity ω . The moment of cutting resistance is proportional to the surface velocity of mowing, Vp and further depends on the properties of the crop, which are the biological and non-technical parameters that we will call K .

All computations will be made in the Maple environment.

DOI: [10.21136/panm.2018.01](https://doi.org/10.21136/panm.2018.01)




Figure 1: Drum mower – Detail of the knife fastening – Coordinate system definition

2. Surface velocity of mowing – fundamentals

First, we start the program and define the basic variables and their design dimensions. All dimensions will be given in the SI-system of units.

Other variables are M the blade weight, RR the carrier drum radius, and V the mower's velocity. Next, we define Mr the transformation matrix for rotation about the angle ψ around the support pin and MR the transformation matrix for rotation about the angle α around the origin of the coordinate system.

```
> restart: with(LinearAlgebra): with(plots):
> Nsu:= [b=0.093, a=0.048, h=0.004, Sb=0.013, r=0.010, M=0.113, R=0.117, RR=0.132,
        omega=50*2*Pi, V=3.0]:
> Mr:=Matrix([[cos(psi), sin(psi)], [-sin(psi), cos(psi)]]):
> MR:=Matrix([[cos(alpha), sin(alpha)], [-sin(alpha), cos(alpha)]]):
```

Now we can determine the coordinates of all vertices of the rectangular blade. First in the basic position corresponding to the position in Fig. 1: ld left bottom corner, lh left upper corner, pd right bottom corner, and ph right top corner. When rotating about the angle ψ around the support pin, these vertices move to Ld , Lh , Pd , and Ph , respectively.

```
> ld:= [-a/2, -Sb]: lh:= [-a/2, -Sb+b]: pd:= [a/2, -Sb]: ph:= [a/2, -Sb+b]:
> Ld:=convert(evalm(ld.Mr+[0,R]),list): Lh:=convert(evalm(lh.Mr+[0,R]),list):
> Pd:=convert(evalm(pd.Mr+[0,R]),list): Ph:=convert(evalm(ph.Mr+[0,R]),list):
```

The cutting edge is given by the intersection of the line joining the Ld and Lh blade corners with a circle with center at the origin of the coordinate system and RR , the radius of the carrier drum. We name this intersection Ldr , the length of the cutting edge then equals the distance of the points Ldr and Lh .

```
> rh:=expand(Ld+p*(Lh-Ld)): eq1:=add(w^2,w=rh)=RR^2: p2:=solve(eq1,p):
> p:=p2[1]: evalf(subs(Nsu,psi=-Pi/12,p2)): Ldr:=rh:
```

Next we compute the position of the blade vertices, LD , LH , PD , PH , and the lower end of the cutting edge LDr , by rotating the system by the angle α around the origin and by shifting it by some $-dx$ in the X direction.

```
> LD:=convert(evalm(Ld.MR),list)+[-dx,0]: LH:=convert(evalm(Lh.MR),list)+[-dx,0]:
> PH:=convert(evalm(Ph.MR),list)+[-dx,0]: PD:=convert(evalm(Pd.MR),list)+[-dx,0]:
> LDr:=convert(evalm(Ldr.MR),list)+[-dx,0]:
```

To check correctness, we plot the position of the blade for $\psi = \pi/12$, $\alpha = 0$, $dx = 0$ in gray and for $\psi = \pi/6$, $\alpha = \pi/24$, $dx = 0.15$ in black. The mowed area is marked in a light gray color, see Fig. 2. We do not list the corresponding Maple commands to save space.




Figure 2: Visualization of the cropped area

3. Surface velocity of mowing – general

Let us assume that, at time $t = 0$, the cutting edge is determined by $[[A1x, A1y], [B1x, B1y]]$. After a short time interval dt , the cutting edge moves to $[[A2x, A2y], [B2x, B2y]]$. If dt is small we can compute $[[A2x, A2y], [B2x, B2y]]$ using Taylor's series and simplify the relationship for the mowing surface velocity VS .

```
> S:=1/2*B1x*B2y+1/2*A1x*B1y-1/2*A2x*B2y-1/2*A1x*A2y-1/2*B2x*B1y
-1/2*B1x*A1y+1/2*A2x*A1y+1/2*A2y*B2x:
> S2:=expand(subs(A1x=A1x(t),A1y=A1y(t),B1x=B1x(t),B1y=B1y(t),
A2x=A1x(t)+diff(A1x(t),t)*dt,A2y=A1y(t)+diff(A1y(t),t)*dt,
B2x=B1x(t)+diff(B1x(t),t)*dt,d=V*dt,B2y=B1y(t)+diff(B1y(t),t)*dt,S)):
> S2:=factor(subs(dt^2=0,S2)): VS:=S2/dt;
```

$$VS := \frac{1}{2} ((A1y(t) - B1y(t)) A1x(t)' - (A1x(t) - B1x(t)) (A1y(t)' + (A1y(t) - B1y(t)) B1x(t)' - (A1x(t) - B1x(t)) B1y(t)') \quad (1)$$

Now we substitute $dx = Vt$, $\alpha = \omega t$ and $\psi = \psi(t)$ in the coordinates $[LDR, LHR]$ of the endpoints of the line segment representing the cutting edge. Furthermore we substitute the points $A1, B1$ in VS by these expressions and receive the final expression of the surface velocity of the mowing Vs .

```
> LDR:=subs(psi=psi(t),dx=V*t,alpha=omega*t,LDr):
> LHR:=subs(psi=psi(t),dx=V*t,alpha=omega*t,LH):
> Vs:=simplify(eval(subs(A1x(t)=LDR[1],A1y(t)=LDR[2],B1x(t)=LHR[1],
B1y(t)=LHR[2],VS))):
```

4. Derivation of the torque equation

4.1. Determining the torque acting on the blade

The calculation of the returning torque is based on the integration of the torque dMz acting on the material element of the blade dm . The position of dm is specified in local coordinates $[\xi, \eta]$, see Fig. 1.

```
> A:=[xi*cos(Pi/2-psi(t)),R+xi*sin(Pi/2-psi(t))]: # dm in local coordinates
> B:=A+[eta*cos(psi(t)),-eta*sin(psi(t))]: # dm in global coordinates
> L:=simplify(sqrt(add(w^2,w=B))): # radius of rotation of dm
> phi:=omega*t:
> dF:=L*omega^2*dm*[sin(phi),cos(phi)]: #elementary centrif. force acting on dm
> su:=[sin(phi)=B[2]/L,cos(phi)=B[1]/L]: # sin and cos substitution
> dFs:=factor(expand(subs(su,dF))): # dF in the local coordinates
```

The torque dMz acting on dm can be expressed as the vector product of the force arm vector $[\xi, \eta]$ and dFs , which can be expressed as the determinant of the matrix Mat .

$$Mat = \begin{vmatrix} [1, 0, 0] & [0, 1, 0] & [0, 0, 1] \\ \xi & \eta & 0 \\ dFs_x & dFs_y & 0 \end{vmatrix}$$

```

> Mat:=Matrix([[1,0,0],[0,1,0],[0,0,1]],[xi,eta,0],[dFs[],0]):
> dMz:=expand(Determinant(Mat)): # elementary torque acting on dM
> dMz:=dMz[3]: # dMz is parallel with the z axis
> Mz:=Int(Int(subs(dM=h*rho,dMz),xi=-Sb..b-Sb),eta=-a/2..a/2); # total torque
> Mz:=simplify(value(Mz),symbolic);

```

$$Mz := \frac{\omega^2 M \sin(\psi(t)) (a^2 + 12 Sb^2 - 12 b Sb + 4 b^2)}{12}$$

4.2. Determining the momentum of the inertia of the blade

The moment of inertia, J , of the blade relative to the axis of rotation around the mounting pin is similar to the calculation of the moment of force in the previous section.

```

> J:=Int(Int((xi^2+eta^2)*h*rho,xi=-Sb..b-Sb),eta=-a/2..a/2):
> J:=simplify(subs(rho=M/(h*a*b),value(J)));

```

$$J := \frac{M(a^2 + 12 Sb^2 - 12 b Sb + 4 b^2)}{12}$$

4.3. The resulting form of the torque equation

The basic form of the torque equation is $\vec{H} = J \vec{\epsilon}$, derived by Newton, where \vec{H} is the torque, J is the momentum of the inertia and $\vec{\epsilon}$ is the vector of the angular acceleration.

The resulting shape of the torque equation De is obtained after substitution J by the expression above and H by Mz and $\vec{\epsilon}$ by $\psi''(t)$:

```

> DE:=J*epsilon=Mz: sue:=epsilon=-diff(psi(t),t,t):
> DE:=subs(sue,DE): De:=diff(psi(t),t,t)=solve(DE,diff(psi(t),t,t));

```

$$De := \psi(t)'' = -\omega^2 \sin(\psi(t)) \quad (2)$$

If we substitute $\omega^2 = g/l$, where g = gravitational acceleration, l = distance of the center of gravity of the mathematical pendulum from the suspension axis, in the equation (2), it will describe the swinging of the mathematical pendulum in Earth's gravitational field. This means that the blade acts as a mathematical pendulum when deflected from the equilibrium. The behavior of the mathematical pendulum is described in many ways, for example [1]. So let's study the behavior of a blade that is deflected from a equilibrium due to mowing resistance.

5. The torque equation with the cutting resistance

Equation (1) describes the surface mowing velocity for a drum with a single blade. In the case of a multi-blade drum, the first blade moves above the surface that has already been cut by another blade. This situation is illustrated in Fig. 3. The problem is that the movement of the second or other blades are on the motion of




Figure 3: Two blades drum – surfaces mowed by the individual blades

the first blade is independent. Therefore deriving the general equation of the actual surface velocity of mowing of the first blade is not possible.

Another problem is that we do not know the cutting resistance of the material to be cut. This means that the torque that acts on the blade is given by the product of an unknown mowing surface velocity and unknown cutting resistance. Given that the product of two unknown values is again an unknown value, it is possible to replace these two unknown quantities with only one unknown quantity and multiply this by the estimated relative change of the mowing surface velocity.

Let's assume that the relative mowing surface velocity can be described by the function: $-\cos(x) \text{Heaviside}(-\cos(x))$, see Fig. 4.




Figure 4: The relative mowing surface velocity

Using this the torque equation considering the cutting resistance becomes:

$$> DE2 := 'J' * 'epsilon' = 'Mz' - K * 'Vs' * \text{Heaviside}(-\cos(\omega t - \psi(t))) * (-\cos(\omega t - \psi(t))) :$$

$$DE2 := J \epsilon = Mz + K Vs \text{Heaviside}(-\cos(\omega t - \psi(t))) \cos(\omega t - \psi(t)) \quad (3)$$

Equation (3) is an extremely nonlinear second order differential equation for $\psi(t)$ the angle of the blade describing the offset from the equilibrium position. The solution can only be obtained using numerical methods.

Because the frequency of the blade vibrations will be similar to the frequency of the rotation of the carrier drum, as indicated by Equation (2), we will only study the blade vibrations as a function of time. For this we use Maple procedure `Fsol`. The input for this procedure is the initial conditions `INI`, the numerical value `K` of the coefficient and the length of the time interval `T`. With these values the `Fsol` procedure computes the numerical solution of the (3) equations using the Runge-Kutta method. This solution is an input of Newton's iteration to find the time at which the blade deflection is maximal. Time and amplitude are stored and plotted later. We do not list the corresponding Maple commands to save space.

```

Fsol:=proc(INI,K,T) local de2, SolN, tau, DT, Q;
  de2:=evalf(subs(Nsu,DE2));
  SolN:=dsolve({de2,INI[]},psi(t),numeric,method=rkf45,maxfun=500000);
  tau:=0.02; DT:=1; Q:=[[0,evalf(rhs(INI[1]))]];
  while tau<=T do;
    while abs(DT)>1e-8 do;
      DT:=evalf(subs(SolN(tau),diff(psi(t),t)/rhs(de2))); tau:=tau-DT;
    end do;
    Q:=[Q[],[tau,rhs(SolN(tau)[2])]]; tau:=2*Q[-1][1]-Q[-2][1]; DT:=1;
  end do;
  map(u->[u[1],evalf(u[2]/Pi*180)],Q):
end proc:

```

With the coefficient `K` equal to 1/100, 2/100 and 3/100 and with the following initial conditions: $\psi(0) = \pi/6$ and $\psi'(0) = 0$ we obtain Fig. 5. It displays the time development of the amplitude of the blade vibrations.




Figure 5: Amplitude of $\psi(t)$ for $\psi(0) = 30^\circ$, $0 \leq t \leq 10$

Now we can use the initial conditions $\psi(0) = 0$ and $\psi'(0) = 0$ and the same values of the coefficient `K`, see Fig. 6.




Figure 6: Amplitude of $\psi(t)$ for $\psi(0) = 0^\circ$, $0 \leq t \leq 10$

6. Conclusion

From the mathematical model presented, it is clear that the blade vibrates around the equilibrium position during mowing. From the graphical results, we see that the frequency of the blade vibrations depends on the mowing surface velocity multiplied by the cut resistance K . The lower this product is, the more the frequency of the blade vibrations approaches the frequency of the carrier drum.

Otherwise, if the product of Vs times K is higher, the frequency of the blade oscillations decreases and it leads to the resonance shocks with the frequency of the carrier drum rotation, which results in a gradual decrease and an increase of the maximum deflection angle of the blade. With increasing cutting resistance, the frequency of resonant shocks increases.

For double blade mowers, this means that the blades need not be balanced with each other relative to the axis of rotation, which can lead to increased stress on the bearings of the carrier drum.

Acknowledgments

The author would like to thank Prof. Dr. Walter Gander for his valuable advice on numerical mathematics and his help in correcting the English version of this article.

References

- [1] Gitterman, M.: Oscillator and Pendulum with a Random Mass. World Scientific Press, Singapore, 2015, 153 pp.
- [2] https://www.poettinger.at/en_uk/Produkte/Detail/50/eurocat-front-mounted-drum-mowers#highlights 30. 8. 2015